LEGENDRE

PURPOSE

Compute the Legendre polynomial of order N, the normalized Legendre polynomial of order N, the normalized associated Legendre polynomial of order N, the normalized Legendre function of the first or second kind, or the normalized associated Legendre function of the first or second kind.

DESCRIPTION

From Abramowitz and Stegum (see REFERENCE below), a system of nth degree polynomials $f_n(x)$ is called orthognal on the interval a <= x <= b with respect to a weight function w(x) if it satisfies the equation:

$$\int_a^b w(x) f_n(x) f_m(x) dx = 0 \qquad m, n = 0, 1, 2, \ldots, (n \neq m) \qquad \text{(EQ Aux-223)}$$

Legendre polynomials use the weight function 1 and are orthogonal for the interval -1 < x < 1. Legendre polynomials can also be defined by the following equation:

$$P_n(x) = \frac{1}{2^n} \sum_{m=0}^{\left[\frac{n}{2}\right]} (-1)^m \binom{n}{m} x^{n-2m} \binom{2n-2m}{n} \qquad \text{(EQ Aux-224)}$$

where [] means the integer portion. Normalized Legendre polynomials are a scaled version of the Legendre polynomial (the scaling is performed so that the integral defined above is equal to 1).

Associated Legendre Polynomials are defined in terms of the standard Legendre polynomial as follows:

$$P_n^m(x) = (-1)^m (1-x^2)^{m/2} \frac{d^m}{dx^m} P_n(x) \qquad \text{(EQ Aux-225)}$$

The value of m defines the degree of the associated Legendre polynomial.

The Legendre functions and associated Legendre functions are generalizations of Legendre polynomials and associated Legendre polynomials where non-integer values of n and m are allowed.

DATAPLOT calculates the standard Legendre polynomial using the following recurrence relation:

$$P_n(x) = x\, P_{n-1}(x) + \left(\frac{n}{n+1}\right)(x\, P_{n-1}(x) - P_{n-2}(x)) \qquad \text{(EQ Aux-226)}$$

where the first few terms for the recuurence were obtained from the Handbook of Mathematical Functions (see the REFERENCE below). The remaining Legendre polynomials and functions are computed using the NORMP set of routines from the Slatec library. These routines use a technique called extended range arithmetic to avoid underflow and overflow problems. However, DATAPLOT stores the result as a single precision real number. If it is unable to do so, it prints an error message.

SYNTAX 1

LET <y> = LEGENDRE(<x>,<n>) <SUBSET/EXCEPT/FOR qualification>

where <x> is a number, parameter, or variable in the range $(-\pi,\pi)$;
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre polynomial;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre polynomial value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the un-normalized Legendre polynomials.

SYNTAX 2

LET <y> = LEGENDRE(<x>,<n>,<m>) <SUBSET/EXCEPT/FOR qualification>

where <x> is a number, parameter, or variable in the range $(-\pi,\pi)$;
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre polynomial;
<m> is a non-negative integer number, parameter, or variable that specifies the degree of the Legendre polynomial;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre polynomial value is stored;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the un-normalized associated Legendre polynomials.

SYNTAX 3

LET <y> = NRMLEG(<x>,<n>) <SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable in the range (-π,π);
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre polynomial;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre polynomial value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the normalized Legendre polynomials.

SYNTAX 4

LET <y> = NRMLEG(<x>,<n>,<m>) <SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable in the range (-π,π);
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre polynomial;
<m> is a non-negative integer number, parameter, or variable that specifies the degree of the Legendre polynomial;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre polynomial value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the normalized associated Legendre polynomials.

SYNTAX 5

LET <y> = LEGP(<x>,<n>) <SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable in the range (-π,π);
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre function;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre function value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the normalized Legendre function of the first kind.

SYNTAX 6

LET <y> = LEGP(<x>,<n>,<m>) <SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable in the range (-π,π);
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre function;
<m> is a non-negative integer number, parameter, or variable that specifies the degree of the Legendre function;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre function value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the normalized associated Legendre function of the first kind.

SYNTAX 7

LET <y> = LEGQ(<x>,<n>) <SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable in the range (-π,π);
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre function;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre function value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the normalized Legendre function of the second kind.

SYNTAX 8

LET <y> = LEGQ(<x>,<n>,<m>) <SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable in the range (-π,π);
<n> is a non-negative integer number, parameter, or variable that specifies the order of the Legendre function;
<m> is a non-negative integer number, parameter, or variable that specifies the degree of the Legendre function;
<y> is a variable or a parameter (depending on what <x> is) where the computed Legendre function value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the normalized associated Legendre function of the second kind.

EXAMPLES

LET A = LEGENDRE(0.5,4)
LET A = LEGENDRE(0.5,4,2)
LET A = NRMLEG(0.5,4)
LET A = NRMLEG(0.5,4,2)
LET A = LEGP(0.5,4,2.5)
LET A = LEGQ(0.5,4,2.5)
LET X2 = LEGENDRE(X1,N)
LET X2 = LEGENDRE(X1,N,A)

NOTE

Legendre polynomials are often specified with an angular input value. That is,

$$P_n(x) = P_n(\cos(\theta))$$

where $-\pi < \theta < \pi$. DATAPLOT uses this convention in calculating the various Legendre polynomials and functions for better accuracy. If your input value is in terms of the (-1,1) interval, use the command

LET XNEW = ARCCOS(X)

and then use XNEW as the input argument to the Legendre functions. By default, the angle is specified in radians. Enter the command DEGREES to specify degree units.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

SPHRHRM	=	Compute the spherical harmonic function.
CHEBT	=	Compute the Chebychev polynomial first kind, order N.
CHEBU	=	Compute the Chebychev polynomial second kind, order N.
HERMITE	=	Compute the Hermite polynomial of order N.
JACOBIPE	=	Compute the Jacobi polynomial of order N.
ULTRASPH	=	Compute the ultrasperical polynomial of order N.
LAGUERRE	=	Compute the Laguerre polynomial of order N.

REFERENCE

"Extended-Range Arithmetic and Normalized Legendre Polynomials," Smith, Olver, and Lozier, ACM Transactions On Mathematical Software, Vol. 7, No. 1, March, 1981 (pp. 93-105).

"Associated Legendre Functions on the Cut," Smith, and Olver, Journal of Computational Physics, Vol. 51, No. 3, September, 1983, (pp. 502-518).

"Handbook of Mathematical Functions, Applied Mathematics Series, Vol. 55," Abramowitz and Stegun, National Bureau of Standards, 1964 (chapter 22).

APPLICATIONS

Mathematics

IMPLEMENTATION DATE

95/7

PROGRAM 1:

```
TITLE CASE ASIS; LABEL CASE ASIS
LINE SOLID DASH DOT DASH2; X1LABEL X; DEGREES
MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
TITLE Legendre Polynomials (order 2 thru 5); Y1LABEL Pn(X)
PLOT LEGENDRE(X,2) FOR X = -179 1 179 AND
PLOT LEGENDRE(X,3) FOR X = -179 1 179 AND
PLOT LEGENDRE(X,4) FOR X = -179 1 179 AND
PLOT LEGENDRE(X,5) FOR X = -179 1 179
.
TITLE Associated Legendre Polynomials (order 2 thru 5)
Y1LABEL P(X,N,M); X2LABEL Degree (M) = 3
PLOT LEGENDRE(X,2,3) FOR X = 1 1 179 AND
PLOT LEGENDRE(X,3,3) FOR X = 1 1 179 AND
PLOT LEGENDRE(X,4,3) FOR X = 1 1 179 AND
PLOT LEGENDRE(X,5,3) FOR X = 1 1 179
.
TITLE Normalized Legendre Polynomials (order 2 thru 5); Y1LABEL Pn(X)
PLOT NRMLEG(X,2) FOR X = -179 1 179 AND
PLOT NRMLEG(X,3) FOR X = -179 1 179 AND
PLOT NRMLEG(X,4) FOR X = -179 1 179 AND
PLOT NRMLEG(X,5) FOR X = -179 1 179
.
TITLE Normalized Associated Legendre Polynomials (order 2 thru 5)
Y1LABEL P(X,N,M); X2LABEL Degree (M) = 3
PLOT NRMLEG(X,2,3) FOR X = -179 1 179 AND
PLOT NRMLEG(X,3,3) FOR X = -179 1 179 AND
PLOT NRMLEG(X,4,3) FOR X = -179 1 179 AND
PLOT NRMLEG(X,5,3) FOR X = -179 1 179
END OF MULTIPLOT
```



PROGRAM 2:

```
TITLE CASE ASIS; LABEL CASE ASIS
LINE SOLID DASH DOT DASH2
DEGREES
MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
TITLE Associated Legendre functions (first kind)
Y1LABEL P(X,N); X1LABEL X; X2LABEL Order (n) = 2.5, 3.5, 4.5, 5.5
PLOT LEGP(X,2.5) FOR X = 1 1 89 AND
PLOT LEGP(X,3.5) FOR X = 1 1 89 AND
PLOT LEGP(X,4.5) FOR X = 1 1 89 AND
PLOT LEGP(X,5.5) FOR X = 1 1 89
TITLE Associated Legendre functions (first kind)
Y1LABEL P(X,N,M); X2LABEL Order (n) = 2.5, 3.5, 4.5, 5.5, Degree (m) = 3
PLOT LEGP(X,2.5,3) FOR X = 1 1 89 AND
PLOT LEGP(X,3.5,3) FOR X = 1 1 89 AND
PLOT LEGP(X,4.5,3) FOR X = 1 1 89 AND
PLOT LEGP(X,5.5,3) FOR X = 1 1 89
TITLE associated Legendre functions (second kind)
Y1LABEL Q(X,n); X2LABEL Order (n) = 2.5, 3.5, 4.5, 5.5
PLOT LEGQ(X,2.5) FOR X = 1 1 89 AND
PLOT LEGQ(X,3.5) FOR X = 1 1 89 AND
PLOT LEGQ(X,4.5) FOR X = 1 1 89 AND
PLOT LEGQ(X,5.5) FOR X = 1 1 89
TITLE Associated Legendre functions (second kind)
Y1LABEL Q(X,N,M); X2LABEL Order (n) = 2.5, 3.5, 4.5, 5.5, Degree (m) = 3
PLOT LEGP(X,2.5,3) FOR X = 1 1 89 AND
PLOT LEGP(X,3.5,3) FOR X = 1 1 89 AND
PLOT LEGP(X,4.5,3) FOR X = 1 1 89 AND
PLOT LEGP(X,5.5,3) FOR X = 1 1 89
END OF MULTIPLOT
```